ISTITUTO NAZIONALE DI FISICA NUCLEARE

Sezione di Bologna

INFN/BE-72/6
23 Ottobre 1972

I. Massa: A COMPUTER PROGRAMME TO DESCRIBE ION MOTION IN SINGLE-CRYSTALS.

Reparto Tipografico
dei Laboratori Nazionali di Frascati

Istituto Nazionale di Fisica Nucleare
Sezione di Bologna

INFN/BE-72/6
23 Ottobre 1972

I. Massa: A COMPUTER PROGRAMME TO DESCRIBE ION MOTION IN SINGLE-CRYSTALS.

This programme can be used in several kinds of studies concerned with the passage of charged ions in single crystals. In the following a version is presented with reference to the axial blocking. This because the programme was originally written as an aid in the interpretation of lifetime measurements, exploiting the axial blocking, made by a Bologna-Harwell group[1]. Anyway the programme structure may be easily adapted to different phenomena under investigation (e. g. channelling radiation damage, flux peaking etc.) and to several kinds of crystal lattices.

Some advantages were found in computer simulations of ions motion in crystals, with respect to analytical techniques[2], mainly as for the possibility of changing easily the general conditions under which phenomena take place.

The ultimate limitation of such a technique is given by the quite long computer times required to obtain an accurate enough statistics, which can make this procedure too expensive.

Description of the model (schematic). -

The programme may deal with three kinds of crystal structures among the most frequently used in the experiments: face-centered cubic (f. c. c.), body-centered cubic (b. c. c.) and diamond. The main axes of each structure were considered, i. e. <100>, <110>, <111>. The programme listing of this report refers to a f. c. c. structure.

Charged ions are supposed to follow classical trajectories[3] within a regulare crystal lattice and to interact with crystal atoms through a central Thomas - Fermi potential. We assume also that the ions have only one important interaction at a time: for each interaction the deflection is computed by the impulse approximation.

Lattice atoms are allowed to oscillate, by thermal vibration, around their equilibrium positions.

As for the blocking, the starting plane of every ion is randomly chosen within the crystal, while the position in this plane is the one of the first scattered atom.

If one is concerned with compound nuclear lifetime effects this position is corrected by a displacement weithted on a suitable exponential position distribution, which depends on the lifetime and on the recoil velocity of the compound system.

The ion emerging direction is randomly chosen within a rotation cone of prefixed width centered around the blocking axis.

More physical details about the model can be found in ref. (4) and in references therein.

The programme classifies the ions leaving the crystal according to their direction with respect to the blocking axis and gives the ions distribution on a plane perpendicular to the blocking axis; beside this it calculates some quantities often used in literature.

Language and speed. -

The programme has been compiled in Fortran IV for use on the CDC 6600 computer at Centro Interuniversitario dell'Italia Nord Orientale (Casalecchio-Bologna). A time of about 365 sec. is required to follow 1000 ions trajectories through a 1000 atomic planes thick crystal (i. e. 1444 Å for an Ag crystal, <110> axis).

The programme. -

In the first page of the listing all the quantities can be found which may be adapted to the particular conditions of interest. In the following the meaning of the non-trivial symbols is explained.

ASSE This parameter as to be fixed equal to 100., 110., or 111. if one is concerned respectively with <100>, <110>, or <111> axis.

NMAX The value of this parameter gives the number of incident ions (this number has to be fixed according to the statistical accuracy required).

NPRO This parameter is connected to the crystal thickness through the following definition:

$$\text{NPRO} = \frac{\text{crystal thickness (Å)}}{\text{DIMZ (Å)}}$$

where DIMZ (see later) denotes the distance, in the crystal, between two adjacent atomic planes perpendicular to the axis. As the first ion-atom collision may take place everywhere within the cry-

stal, the number of the atoms which are taken into account in a single trajectory is between 1 and NPRO.

NOSC The value of this parameter gives the number of thermally vibrating atoms along each trajectory. In the programme these NOSC vibrating atoms are adjacent and ordered starting from the atom with which the first collision takes place.

APER This parameter gives the half width (in rad.) of the rotation cone where ions can be emitted.

RANGX } The range of every atomic potential is assumed to be a rectangle
RANGY } on a plane perpendicular to the row. RANGX and RANGY (in Å) measure the sides of this rectangle.

TAU In lifetime effect studies this parameter is given by the product $v_\perp \tau$ (in Å), where τ is the mean lifetime of the compound system and $v_\perp$ is the component of the recoil velocity normal to the row. In absence of lifetime effects TAU has to be set equal to 0.

SCAT } With reference to Figs. 4 and 5, the values of these parameters
SCAX } denote the dimensions of the intervals where ions leaving the crystal are classified.

The programme works in terms of a polar reference frame, whose axis lies along the row. The sense in which the ion moves into the crystal is assumed as positive. With reference to Fig. 1, θ_X and θ_Y denote the "components" of θ on a plane normal to the row. In the low-angle approximation (which is usually widely satisfied)

(1) $$\theta^2 = \theta_X^2 + \theta_Y^2 \qquad \varphi = \mathrm{tg}^{-1}\left(\frac{\theta_Y}{\theta_X}\right)$$

Fig.1



FIG. 1 - Reference frame to follow the ions through the crystal.

Ions leaving the crystal are classified in two ways: according to their θ (i. e. without reference to φ) and according to their $[\theta_X, \theta_Y]$ respectively. The programme computes the values of the corresponding intervals, according to the chosen dimensions (SCAT, SCAX). Successivel y the programme computes the Thomas-Fermi screening parameter

$$a_{TF} = 0.8853 \frac{\hbar^2}{m e^2} \left[Z_1^{2/3} + Z_2^{2/3} \right]^{-1/2} \tag{2}$$

where $\hbar$ Planck constant/2π
m, e electron mass and charge
Z_1, Z_2 atomic numbers of colliding particles

and the root mean square displacement due to thermal vibrations, in each direction

$$\overline{x^2} = \frac{3\hbar^2 T}{M k \Theta^2} \left[1 + \frac{\Theta^2}{36 T^2} \right] \tag{3}$$

where k Boltzmann constant
M crystal atomic mass, in a. m. u.
Θ crystal Debye temperature, in ^{o}K
T crystal temperature, in ^{o}K.

It follows that the position probability density of an atom along a direction (e. g. X) may be written as

$$P(x) = \frac{\exp\left[-x^2/2\overline{x^2}\right]}{\sqrt{2\pi\overline{x^2}}} \tag{4}$$

An exponential law is assumed in the case TAU $\neq$ 0 for the decay of the nuclear compound states

$$P(s) = \frac{1}{\bar{s}} \exp\left[-s/\bar{s}\right] \tag{5}$$

where s is the distance of the compound nucleus from the row and $\bar{s} = v_{\perp}\tau \equiv$ $\equiv$ TAU. The programme now is ready to follow the first ion through the crystal. The emission direction of the ion is randomly chosen within a cone (see the definition of APER) by means of the subroutine RANF which generates random numbers between 0 and 1.

If TAU = 0 the emission position is temporarely set in a point of the row; if TAU $\neq$ 0 this position is obtained from the exponential distribution 5.

The number of atomic planes along the path of the ion is then randomly chosen between 0 and NPRO.

The emission position is then corrected for thermal vibration by means of the distribution 4, following the method of ref. (5).

The ion maintains its direction up to the next atomic plane, where its position with respect to the row is computed. With reference to Fig. 2 X1 is set equal to 0.



$$BX = XD + DBX + TETX * DIMZ - X1$$

FIG. 2 - How the programme computes the ion-atom impact parameter (each dimension).

If the atom with which the ion interacts is allowed to vibrate X1 is computed as outlined above (the same procedure holds for the Y direction). For each ion - atom interaction a translation of the reference frame is performed, displacing the origin on the row to which the colliding atom belongs. Of course this translation depends on the crystallographic axis involved. Fig. 3 shows, for instance, the structure corresponding to an <110> axis: along the Z axis, planes with atoms o alternate with planes with atoms ●.

In this figure the way to choose the atom interacting with the ion is also shown: let us suppose that the ion was coming from a plane ● and it is now in a plane o; as a first step the programme computes to what dashed rectangle the ion belongs, and its impact parameter with respect to the corresponding atom.

If the ion is inside the range of the atomic potential (as determined by RANGX and RANGY) the deflection is computed by the impulse approximation, assuming for the Thomas - Fermi potential the expansion given by Molière:



FIG. 3 - Maximum atomic ranges in a plane perpendicular to the crystallographic axis.

$$(6) \qquad \Delta\vartheta_X = \frac{Z_1 Z_2 e^2}{E\, a_{TF}} \frac{b_x}{b} \left[0.6\, K_1\left(\frac{6b}{a_{TF}}\right) + 0.66\, K_1\left(\frac{1.2\,b}{a_{TF}}\right) + 0.105\, K_1\left(\frac{3b}{a_{TF}}\right) \right]$$

where b, b_x impact parameter and its projection on X axis
E ion energy, in the centre of mass frame
$K_1(x)$ modified Bessel functions of order one.

A similar formula holds for $\Delta\theta_y$. Bessel functions are computed by the subroutine BESK.

In this way a new set of θ_x, θ_y, θ and φ is obtained and this procedure is repeated for each atomic plane crossed by the ion.

As pointed out above, ions leaving the crystal are classified in two ways: i) according to the values of $[\theta_x, \theta_y]$, ii) according to the values of θ (in this classification ions corresponding to positive and negative θ_x values are separately considered).

The ion distribution in θ is then normalized to unit solid angle. The last quantities computed by the programme (which may be used to some comparison with literature quoted data) are:

a) the Lindhard's critical angle

$$\psi_1 = \sqrt{\frac{2 Z_1 Z_2 e^2}{d E}} \tag{7}$$

where d is the distance between the axis atoms.

b) The Lindhard's half width at half maximum of the blocking dip

$$\psi_{1/2} = \psi_1 \left\{ \frac{1}{2} \ln \left[\frac{C^2 a_{TF}^2}{2 \overline{x^2} \ln 2} + 1 \right] \right\}^{1/2} \tag{8}$$

where C is a constant in the range 1 - 2.

Page I of the print-out shows the ions distribution on a plane normal to the row, which is understood at the centre of the square. The printed numbers represent the contents of square intervals of side SCAX.

In Fig. 4 the relationship is shown between the crystal axes and the reference frame of Fig. 1.

If TAU $\neq$ 0 the programme assumes the displacement of the compound system to be in the positive sense of the X axis. This is why in the classification according to θ, contributions due to positive and negative θ_x values are separately considered (Fig. 5). The angular width of the θ intervals is SCAT.

Page II of the print-out collects the contents of the 72 θ intervals, ordered according to the numbers shown in Fig. 5. Such a distribution can be easily reduced to the usual representation (see Fig. 6).

Page III of the print-out gives the normalized θ distribution.



FIG. 4 - Relationship between crystallographic axes and reference axes as assumed in the print-out representation of the $[\theta_x, \theta_y]$ classification.



FIG. 5 - Classification according to θ. The position of the crystallographic axes is the same as in Fig. 4.



FIG. 6 - A typical blocking dip, as obtained from the θ distribution of page II of the print-out.

In the last page of the print-out the values of the Thomas-Fermi screening length and the one dimentional root mean square vibration amplitude are collected, together with the values of the Lindhard's critical angle and half width at half maximum.

The programme is provided of some internal checks to test if the assumptions (e. g. impulse approximation) are fulfilled by the particular conditions under investigation.

REFERENCES. -

(1) - G. J. Clark et al., Nuclear Phys. A173, 73 (1971).
(2) - J. U. Andersen and L. C. Feldman, Phys. Rev. B1, 2063 (1970).
(3) - J. Lindhard, Mat. Fys. Medd. 34, N. 14 (1965).
(4) - I. Massa, Report INFN/BE-69/9.
(5) - Mathematical Methods for Digital Computers, Edited by A. Ralston and H. S. Wilf (Wiley, 1960) Vol. 2, p. 249.

```
      DIMENSION TETU(72),TETU1(30),K(30,30),K1(144),K2(144)
      REAL      MASSA , LATO                                          PAGE     1
C
C*****************************************************************************
C     PARAMETERS
C*****************************************************************************
C
C     CRYSTALLOGRAPHIC AXIS
      ASSE=110.
C
C     NUMBER OF IONS TRACKED THROUGH THE LATTICE
      NMAX=10000
C
C     MAXIMUM NUMBER OF CROSSED PLANES
      NPRO=1000
C
C     NUMBER OF VIBRATING LATTICE ATOMS , FOR EACH ION
      NOSC=100
C
C     ION MAXIMUM EMISSION ANGLE ( RAD )
      APER=0.018
C
C     CRYSTAL TEMPERATURE ( K )
      TEMP=293.
C
C     DEBYE CRYSTAL TEMPERATURE ( K )
      TDEB=221.
C
C     MASS OF THE LATTICE ATOMS (AMU )
      MASSA=107.
C
C     ATOMIC NUMBER OF THE ION
      Z1=1.
C
C     ATOMIC NUMBER OF THE ATOMS
      Z2=47.
C
C     ION ENERGY IN THE CENTRE OF MASS FRAME ( MEV )
      ERG=5.
C
C     CUBIC LATTICE CONSTANT ( ANGSTROM )
      LATO=4.086
C
C     RANGE OF THE ATOMIC POTENTIAL ( ANGSTROM )
      RANGX=LATO/SQRT(8.)
      RANGY=LATO/2.
C
C     ATOMIC DISTANCE ALONG THE AXIS ( ANGSTROM )
      DILI=LATO/SQRT(2.)
C
C     MEAN DISPLACEMENT OF THE COMPOUND SYSTEM ( ANGSTROM )
      TAU=0.
C
C     INTERVALS DIMENSION FOR TETA CLASSIFICATION ( RAD )
      SCAT=0.001
```

```
C
C       INTERVALS DIMENSION FOR TETAX AND TETAY CLASSIFICATION ( RAD ) PAGE      2
        SCAX=0.0015
C
C*************************************************************************
C
        BART=Z1*Z2/ERG
        IF (BART.GT.3333.) GO TO 42
        IF (APER.GT.0.09) GO TO 44
        NFUO=0
        IF (ASSE.EQ.100.) ASSIGN 220 TO NASSE
        IF (ASSE.EQ.110.) ASSIGN 220 TO NASSE
        IF (ASSE.EQ.111.) ASSIGN 320 TO NASSE
C
C       COMPUTE THE POSITION OF THE TETA INTERVALS
        TETU(1)=SCAT
        DO 1 J=2,72
        L=J-1
    1   TETU(J)=TETU(L)+SCAT
C
C       COMPUTE THE POSITION OF THE TETAX AND TETAY INTERVALS
        TETU1(1)=-14.*SCAX
        DO 4 J=2,30
        L=J-1
    4   TETU1(J)=TETU1(L)+SCAX
C
C       PUT EQUAL TO 0 THE CONTENTS OF ALL THE INTERVALS
        DO 2 J=1,30
        DO 2 I=1,30
    2   K(J,I)=0
        DO 36 J=1,144
        K1(J)=0
   36   K2(J)=0
C
C       COMPUTE THE THOMAS-FERMI SCREENING LENGTH
        ATF=0.8853*0.529*(Z1**(2./3.)+Z2**(2./3.))**(-0.5)
C
C       COMPUTE A ONE DIMENSIONAL ROOT MEAN SQUARE VIBRATION AMPLITUDE
        SIG1=SQRT((2.417*TEMP)*(1.+(TDEB/TEMP)**2/36.)/(MASSA*1.66*(TDEB)*
      1*2))*10.
        NE=1
C
C       RANDOMLY CHOOSE FI BETWEEN 0 AND 6.28318
    5   ABC=RANF(0.)
        FI=6.28318*ABC
C
C       RANDOMLY CHOOSE COS(TETA) BETWEEN COS(APER) AND COS(0)
        ABC=RANF(0.)
        COTE=ABC*(1.-COS(APER))+COS(APER)
        SETE=SQRT(1.-COTE**2)
C
C       COMPUTE TETAX AND TETAY
        TETX=COS(FI)*SETE/COTE
        TETY=SIN(FI)*SETE/COTE
C
```

```
C      DISPLACEMENT FROM THE AXIS FOR TAU EQUAL TO 0
       DBX=0.
       DBY=0.
C
C      DISPLACEMENT FROM THE AXIS FOR TAU NOT EQUAL TO 0
       ABC=RANF(0.)
       DBX=-TAU*ALOG(ABC)
C
C      RANDOMLY CHOOSE THE EMISSION PLANE BETWEEN 0 AND NPRO
   34  ABC=RANF(0.)
       NA=INT(ABC*FLOAT(NPRO))
C
C      RANDOMLY CHOOSE THE POSITION OF THE EMITTING ATOM (THERMAL VIBRATION)
       ABC=RANF(0.)
       ABC1=RANF(0.)
       X1D=6.2831853*ABC1
       Y1D=SQRT(-2.*ALOG(ABC))
       XD=SIG1*Y1D*COS(X1D)
       YD=SIG1*Y1D*SIN(X1D)
       NNA=1
C
C      COMPUTE THE IMPACT PARAMETER ON THE NEXT PLANE
       IF (ASSE-110.) 230,231,232
C
C*****************************************************************************
C           FACE CENTRED CUBIC STRUCTURE
C*****************************************************************************
C
  230  DIMX=LATO/SQRT(8.)
       DIMY=DIMX
       DIMZ=LATO/2.
       GO TO 210
  231  DIMX=LATO/SQRT(8.)
       DIMY=LATO/2.
       DIMZ=DIMX
       GO TO 210
  232  DIMX=LATO/SQRT(2.)
       DIMY=LATO*SQRT(3./8.)
       DIMZ=LATO/SQRT(3.)
  210  DDIMX=2.*DIMX
       DDIMY=2.*DIMY
       TDIMX=3.*DIMX
       TDIMY=3.*DIMY
       QDIMX=4.*DIMX
       QDIMY=4.*DIMY
       IF (RANGX.GT.DIMX.OR.RANGY.GT.DIMY) GO TO 41
  200  X1=0.
       Y1=0.
       BX=DIMZ*TETX-X1+XD+DBX
       BY=DIMZ*TETY-Y1+YD+DBY
       B=SQRT(BX**2+BY**2)
       GO TO NASSE(220,320)
  220  IF (ABS(BX).GT.QDIMX.OR.ABS(BY).GT.QDIMY) GO TO 5
       IF (BY.LE.(-DDIMY)) GO TO 201
       IF (BY.LE.0.) GO TO 202
```

```
      IF (BY.LE.DDIMY) GO TO 203                                        PAGE      4
      BY=BY-TDIMY
      GO TO 204
  201 BY=BY+TDIMY
      GO TO 204
  202 BY=BY+DIMY
      GO TO 204
  203 BY=BY-DIMY
  204 IF(BX.LE.(-DDIMX)) GO TO 205
      IF (BX.LE.0.) GO TO 206
      IF (BX.LE.DDIMX) GO TO 207
      BX=BX-TDIMX
      GO TO 208
  205 BX=BX+TDIMX
      GO TO 208
  206 BX=BX+DIMX
      GO TO 208
  207 BX=BX-DIMX
      GO TO 208
  320 BY=BY-LATO/SQRT(24.)
      IF (ABS(BX).GT.QDIMX.OR.ABS(BY).GT.QDIMY) GO TO 5
      IF (BY.LE.(-DDIMY)) GO TO 301
      IF (BY.LE.0.) GO TO 202
      IF (BY.LE.DDIMY) GO TO 303
      BY=BY-TDIMY
      GO TO 204
  301 BY=BY+TDIMY
      GO TO 304
  303 BY=BY-DIMY
  304 IF (BX.LE.(-TDIMX)) GO TO 306
      IF (BX.LE.(-DIMX)) GO TO 307
      IF (BX.LE.DIMX) GO TO 208
      IF (BX.LE.TDIMX) GO TO 309
      BX=BX-QDIMX
      GO TO 208
  306 BX=BX+QDIMX
      GO TO 208
  307 BX=BX+DDIMX
      GO TO 208
  309 BX=BX-DDIMX
  208 B=SQRT(BX**2+BY**2)
C
C*****************************************************************************
C
      IF (ABS(BX).LE.RANGX.AND.ABS(BY).LE.RANGY) GO TO 13
C
C     OUT OF THE RANGE NO CHANGE IN TETAX AND TETAY
      DTETX=0.
      DTETY=0.
      GO TO 19
   13 NNA=NNA+1
      IF (NNA.GT.NOSC) GO TO 14
C
C     MODIFIE  THE IMPACT PARAMETER FOR THE THERMAL VIBRATION
      ADC=RANF(0.)
```

```
      ABC1=RANF(0.)
      X1D=6.2831853*ABC1
      Y1D=SQRT(-2.*ALOG(ABC))
      X1=SIG1*Y1D*COS(X1D)
      Y1=SIG1*Y1D*SIN(X1D)
      BX=BX-X1
      BY=BY-Y1
      B=SQRT(BX**2+BY**2)
C
C     COMPUTE THE DEFLECTION
   14 COSTA=14.4E-6*Z1*Z2/(ERG*ATF)
      N=1
      XY=6.*B/ATF
      CALL BESK(XY,N,BK,IER)
      BE1=BK
      XY=1.2*B/ATF
      CALL BESK(XY,N,BK,IER)
      BE2=BK
      XY=3.*B/ATF
      CALL BESK(XY,N,BK,IER)
      BE3=BK
      BESSEL=0.6*BE1+0.66*BE2+0.105*BE3
      DTETX=COSTA*BESSEL*BX/B
      DTETY=COSTA*BESSEL*BY/B
C
C     COMPUTE THE NEW TETA , TETAX , TETAY , FI
   19 TETX=TETX+DTETX
      TETY=TETY+DTETY
      TETA=SQRT(TETX**2+TETY**2)
      FI=ATAN(TETY/TETX)
      IF (TETA.LE.0.09) GO TO 43
      NFUO=NFUO+1
      GO TO 21
   43 NA=NA+1
      IF(NA.GT.NPRO) GO TO 20
      XD=X1
      YD=Y1
      DBX=BX
      DBY=BY
      GO TO 200
   20 IF(TETA.GT.TETU(72)) GO TO 21
      IF (ABS(TETX).GT.TETU1(30).OR.ABS(TETY).GT.TETU1(30)) GO TO 22
C
C     CLASSIFIE  THE ION ACCORDING TO THE OUTPUT ANGLE
      DO 24 M1=1,30
      IF (TETX.LE.TETU1(M1)) GO TO 25
   24 CONTINUE
   25 DO 26 M2=1,30
      IF (TETY.LE.TETU1(M2)) GO TO 27
   26 CONTINUE
   27 K(M1,M2)=K(M1,M2)+1
   22 DO 28 N=1,72
      IF(TETA.LE.TETU(N)) GO TO 29
   28 CONTINUE
   29 IF(TETX.LE.0.) N=73-N
```

```
      IF(TETX.GE.0.) N=72+N
      K1(N)=K1(N)+1
   21 NE=NE+1
      IF (NE.LE.NMAX) GO TO 5
C
C     NORMALIZE TETA DISTRIBUTION
      AK=3.1416*TETU(1)/2.
      K2(72)=INT(FLOAT(K1(72))/AK)
      K2(73)=INT(FLOAT(K1(73))/AK)
      DO 32 J=2,72
      L=J-1
      AK=3.1416*(TETU(J)+TETU(L))/2.
      N1=73-J
      N2=72+J
      K2(N1)=INT(FLOAT(K1(N1))/AK)
   32 K2(N2)=INT(FLOAT(K1(N2))/AK)
C
C     COMPUTE THE LINDHARD S CRITICAL ANGLE
      PSI1=SQRT(Z1*Z2*39.464/(DILI*ERG*137.))*10.
C
C     COMPUTE THE HWHM
      PSIM=PSI1*SQRT(0.5*ALOG(3.*ATF**2/((SIG1*1.4142)**2*ALOG(2.))+1.))
C
C     PRINT
      PRINT 100
      PRINT 102
      DO 31 I=5,26
      II=31-I
   31 PRINT 103 , (K(J,II),J=5,26)
      PRINT 100
      PRINT 104
      PRINT 105 , (K1(J),J=1,144)
      PRINT 100
      PRINT 106
      PRINT 105 , (K2(J),J=1,144)
      PRINT 100
      PRINT 107 , ATF
      PRINT 110 , SIG1
      PRINT 111 , PSI1
      PRINT 112 , PSIM
      IF (NFUO.NE.0) PRINT 115 , NFUO
      GO TO 40
   41 PRINT 113
      GO TO 40
   42 PRINT 114
      GO TO 40
   44 PRINT 116
   40 CONTINUE
      STOP
  100 FORMAT (1H1,//)
  102 FORMAT (15X,*TWO DIMENSIONAL DISTRIBUTION OF THE IONS*)
  103 FORMAT (/,2X,30I4)
  104 FORMAT (//////,35X,*TETA ANGULAR DISTRIBUTION*,/////)
  105 FORMAT (20X,10I7,/)
  106 FORMAT (//,35X,*NORMALIZED TETA ANGULAR DISTRIBUTION*,///////)
```

```
107 FORMAT (//////,10X,*THOMAS-FERMI SCREENING LENGTH*,5X,F9.5,3X,*ANG
   1STROM*)
110 FORMAT (//////,10X,*ONE DIMENSIONAL ROOT MEAN SQUARE VIBRATION AMP
   1LITUDE*,5X,F10.6,3X,*ANGSTROM*)
111 FORMAT (//////,10X,*LINDHARD S CRITICAL ANGLE*,5X,F10.5,3X,*MILLIR
   1AD*)
112 FORMAT (//////,10X,*HALF WIDTH AT HALF MAXIMUM*,5X,F11.5,3X,*MILLI
   1RAD*)
113 FORMAT (1H1,/////,10X,*THE RANGES OF ADJACENT ATOMS ARE SUPERIMPOS
   1ED ONE UPON THE OTHER*)
114 FORMAT (1H1,/////,10X,*IMPULSE APPROXIMATION IS NOT ACCURATE ENOUG
   1H*)
115 FORMAT (/////,10X,I5,*  IONS HAVE BEEN TAKEN OFF BECAUSE TETA WAS
   1GREATER THAN 5 DEGREES*)
116 FORMAT (1H1,/////,10X,*THE ANGULAR WIDTH IS TOO BIG FOR THE LOW AN
   1GLE APPROXIMATION*)
    END
```

TWO DIMENSIONAL DISTRIBUTION OF THE IONS

1	4	0	0	9	23	33	21	18	32	13	10	30	15	25	24	28	7	0	2	1	0
1	2	9	5	6	21	28	25	21	25	15	18	33	14	26	36	18	6	5	11	2	2
1	5	14	33	9	9	29	40	21	28	36	19	26	33	23	24	12	6	34	28	6	1
3	12	30	32	8	4	15	25	25	27	21	11	22	23	33	13	4	14	24	29	20	3
12	16	30	33	30	16	4	20	27	23	17	20	22	29	25	6	16	46	28	21	10	17
18	25	24	25	29	43	8	7	35	26	20	19	25	31	6	7	19	24	18	22	31	9
10	9	21	25	26	33	24	9	16	28	19	15	22	14	12	22	35	24	24	23	14	15
27	23	19	11	19	28	29	18	9	17	18	12	22	6	29	28	24	30	11	34	35	18
17	18	31	23	23	13	24	32	20	6	11	11	7	15	30	13	21	27	27	35	20	18
28	44	25	20	34	24	28	18	17	6	0	3	3	10	21	32	28	31	27	31	26	25
7	8	12	12	14	15	22	23	14	2	1	0	2	5	25	24	10	15	12	16	9	12
8	13	12	15	18	17	22	13	11	2	3	0	1	6	21	12	24	17	15	14	9	10
27	32	26	27	23	29	23	25	18	6	4	2	9	9	23	20	44	37	37	27	32	24
20	25	26	32	24	14	28	26	16	15	12	16	5	19	28	27	14	26	34	26	17	25
18	25	24	19	16	18	27	32	10	16	19	16	15	8	27	29	17	24	18	14	35	18
15	6	21	25	30	35	29	8	13	29	21	11	40	23	10	22	23	28	22	20	13	20
18	33	25	26	26	30	6	10	36	24	19	20	19	38	10	11	34	24	23	21	18	8
13	18	20	23	26	16	5	21	25	27	24	18	34	26	26	6	16	30	25	20	18	7
2	14	22	22	21	2	7	30	26	21	22	16	28	23	31	12	9	27	33	25	21	1
0	3	25	25	8	5	26	26	27	21	16	20	27	16	27	31	5	5	29	20	4	0
0	1	11	5	4	19	37	25	17	34	14	18	27	18	23	32	15	6	4	7	2	0
1	1	1	0	5	18	26	27	14	27	10	14	17	18	25	22	32	11	0	1	1	0

TETA ANGULAR DISTRIBUTION

0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	1	0	0	0
0	0	0	0	1	1	0	1	0	1
1	0	0	0	1	0	2	0	0	2
1	0	4	5	3	2	3	6	8	11
21	27	70	249	463	479	431	432	434	388
366	299	296	262	228	195	152	94	36	8
4	1	0	0	13	26	73	145	171	239
286	275	336	361	383	425	451	454	443	511
257	59	37	11	14	2	4	4	2	4
0	1	2	3	0	1	0	1	1	0
2	0	1	0	1	0	0	0	0	1
1	0	0	0	0	0	0	2	0	0
0	0	1	0	0	0	0	0	0	0
1	0	0	0						

NORMALIZED TETA ANGULAR DISTRIBUTION

0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	5	0	0	0
0	0	0	0	6	6	0	7	0	7
7	0	0	0	8	0	17	0	0	19
10	0	43	55	34	24	37	77	108	155
310	419	1142	4284	8421	9240	8851	9483	10233	9880
10130	9064	9917	9811	9676	9549	8796	6649	3274	1018
848	636	0	0	1655	2364	5163	8391	8373	10143
10710	9214	10185	9992	9752	10020	9900	9323	8546	9294
4421	963	574	162	198	27	51	49	24	46
0	10	20	30	0	9	0	8	8	0
16	0	7	0	7	0	0	0	0	6
6	0	0	0	0	0	0	11	0	0
0	0	5	0	0	0	0	0	0	0
4	0	0	0						

THOMAS-FERMI SCREENING LENGTH .12506 ANGSTROM

ONE DIMENSIONAL ROOT MEAN SQUARE VIBRATION AMPLITUDE .091062 ANGSTROM

LINDHARD S CRITICAL ANGLE 9.68083 MILLIRAD

HALF WIDTH AT HALF MAXIMUM 8.72785 MILLIRAD

3 IONS HAVE BEEN TAKEN OFF BECAUSE TETA WAS GREATER THAN 5 DEGREES